



# HW#4 Problem Explanations

ECE304  
Fall 2009



**Notice:**

**Homework 4 due next Fri.  
(Oct. 9)**

- Problem 1, Circuit 1: (a). first determine  $V_{in}$  ( $V_{in} = V_{out}$ )

- For NMOS:

$$V_{gs,1} = V_{ds,1} = V_{in}$$


$$I_{D,1} = \frac{1}{2} \mu_n \frac{\epsilon_{ox}}{t_{ox,n}} (V_{GS,1} - V_{th,n})^2$$

- For PMOS:


$$V_{gs,2} = V_{ds,2} = V_{in} - V_{dd}$$

$$I_{D,2} = \frac{1}{2} \mu_p \frac{\epsilon_{ox}}{t_{ox,p}} (V_{GS,2} - V_{th,p})^2$$

$$I_{D,1} = I_{D,2} \Rightarrow V_{in}$$



■ (a):



Equivalent  
circuit


$$g_m = \mu C_{ox} \frac{W}{L} (V_{GS} - V_{th})$$

$$r_o = \frac{1}{\lambda I_D}$$

$$Av = -(g_{m1} + g_{m2})(r_{o1} // r_{o2})$$

$$R_{out} = r_{o1} // r_{o2}$$

■ (b):



Equivalent  
circuit

$$V_g = V_{in} = 0, \quad V_2 = V_{dd}$$

$$g_{m1}V_g - g_{m2}V_2 + \frac{V_o}{r_{o1}} + \frac{V_o - V_{dd}}{r_{o2}} = 0$$

$$\Rightarrow \frac{V_o}{r_{o1}} + \frac{V_o - V_{dd}}{r_{o2}} - g_{m2}V_{dd} = 0$$

$$\Rightarrow A_v = \frac{V_o}{V_{dd}} =$$

## ■ Circuit 1 Netlist:

```
.title hw4, 1: circuit a

.OPTION CONVERGE=4
.op

.model nch nmos level=1 vto=0.7 gamma=0.5 phi=0.9 nsub=9e+14 ld=0.08e-6 u0=350 lambda=0.1
+tox=9e-9 pb=0.9 cj=0.56e-3 cjsw=0.35e-11 mj=0.45 mjsw=0.2 cgdo=0.4e-9 js=1.0e-8

.model pch pmos level=1 vto=-0.8 gamma=0.4 phi=0.8 nsub=5e+14 ld=0.09e-6 u0=100 lambda=0.2
+tox=9e-9 pb=0.9 cj=0.94e-3 cjsw=0.32e-11 mj=0.5 mjsw=0.3 cgdo=0.3e-9 js=0.5e-8

vdd vdd 0 3

m1 vout vin 0 0 nch w=50u l=0.5u
m2 vout vin vdd 0 pch w=50u l=0.5u

vin vin 0 1.254

.probe vout = v(vout)

.tf v(vout) vin ← Av = Vout / Vin
.alter
.tf v(vout) vdd ← Av = Vout / Vdd

.end
```


- Run simulation and check xxx.lis file (using WINSCP)

**hspice xxx.sp > xxx.lis**


```
/home/ece304/Hw_4/problem1_ckt_a.lis - ece304@ece3.ece.arizona.edu  
**** small-signal transfer characteristics  
v(vout)/vin = -9.2886  
input resistance at vin = 1.000e+20  
output resistance at v(vout) = 391.6797  
  
**** job concluded  
1***** HSPICE -- C-2009.03 32-BIT (Feb 24 2009) sunos *****  
*****  
.title hw4, l: circuit a  
  
***** job statistics summary tnom= 25.000 temp= 25.000 *****  
Line: 1/458 Column: 1 Character: 85 (0x55)
```

```
/home/ece304/Hw_4/problem1_ckt_a.lis - ece304@ece3.ece.arizona.edu  
**** small-signal transfer characteristics  
v(vout)/vdd = 4.4321  
input resistance at vdd = 4.637e-23  
output resistance at v(vout) = 391.6797  
  
**** job concluded  
1***** HSPICE -- C-2009.03 32-BIT (Feb 24 2009) sunos *****  
*****  
.title hw4, l: circuit a  
  
***** job statistics summary tnom= 25.000 temp= 25.000 *****  
***** HSPICE Threads Information *****  
Line: 1/458 Column: 1 Character: 85 (0x55)
```

## ■ Circuit 2: (a). $A_v$ :



## Equivalent circuit




$$V_g = V_{in} \frac{R_2}{R_1 + R_2} + V_o \frac{R_1}{R_1 + R_2}$$

$$\frac{V_o - V_{in}}{R_1 + R_2} + \frac{V_o}{r_{o1} + r_{o2}} + (g_{m1} + g_{m2})V_g = 0$$

$$\Rightarrow A_v = \frac{V_o}{V_{in}} =$$

■ Circuit 2: (a).  $R_{out}$ :




Equivalent  
circuit

$$V_{in} = 0 \Rightarrow V_g = V_o \frac{R_1}{R_1 + R_2}$$

$$i_o = \frac{V_o}{R_1 + R_2} + \frac{V_o}{r_{o1} + r_{o2}} + (g_{m1} + g_{m2})V_o \frac{R_1}{R_1 + R_2} = 0$$

$$\Rightarrow r_o = \frac{V_o}{i_o} =$$

■ (b).



Equivalent  
circuit

$$V_g = V_o \frac{R_1}{R_1 + R_2}, \quad V_2 = V_{dd} - V_g = V_{dd} - V_o \frac{R_1}{R_1 + R_2}$$

$$g_{m1} V_g - g_{m2} V_2 + \frac{V_o}{r_{o1}} + \frac{V_o - V_{dd}}{r_{o2}} + \frac{V_o}{R_1 + R_2} = 0$$

$$\Rightarrow \frac{V_o}{r_{o1}} + \frac{V_o - V_{dd}}{r_{o2}} + \frac{V_o}{R_1 + R_2} + g_{m1} \left( V_o \frac{R_1}{R_1 + R_2} \right) - g_{m2} \left( V_{dd} - V_o \frac{R_1}{R_1 + R_2} \right) = 0$$

$$\Rightarrow A_v = \frac{V_o}{V_{dd}} =$$

## ■ Circuit 2 Netlist:

```
.title hw4, 1: circuit b

.OPTION CONVERGE=4
.op

.model nch nmos level=1 vto=0.7 gamma=0.5 phi=0.9 nsub=9e+14 ld=0.08e-6 u0=350 lambda=0.1
+tox=9e-9 pb=0.9 cj=0.56e-3 cjsw=0.35e-11 mj=0.45 mjsw=0.2 cgdo=0.4e-9 js=1.0e-8

.model pch pmos level=1 vto=-0.8 gamma=0.4 phi=0.8 nsub=5e+14 ld=0.09e-6 u0=100 lambda=0.2
+tox=9e-9 pb=0.9 cj=0.94e-3 cjsw=0.32e-11 mj=0.5 mjsw=0.3 cgdo=0.3e-9 js=0.5e-8

vdd vdd 0 3

m1 vout x 0 0 nch w=50u l=0.5u
m2 vout x vdd 0 pch w=50u l=0.5u
r1 vin x 1k
r2 x vout 10k


vin vin 0 1.254

.probe vout = v(vout)

.tf v(vout) vin
.alter
.tf v(vout) vdd

.end
```

## ■ Problem 2, (a): calculate $V_{DS}$



$$V_{o1} = V_{o2} = V_{DD} - \frac{I_{SS}}{2} R_D = 1.1 \text{ V}$$

$$V_{DSAT} = \sqrt{\frac{2I_{DS}}{k' \frac{W}{L}}} = \sqrt{\frac{I_{SS}}{k' \frac{W}{L}}} = 0.1 \text{ V}$$

$$V_S = V_{ic} - V_T - V_{DSAT}$$

$$= V_{ic} - (V_{T0} + \gamma(\sqrt{2\phi + V_S} - \sqrt{2\phi})) - V_{DSAT}$$

$$= 0.9 - (0.5 + 0.5(\sqrt{0.6 + V_S} - \sqrt{0.6})) - 0.1$$

$$= 0.3 - 0.5(\sqrt{0.6 + V_S} - \sqrt{0.6})$$

Solve iteratively to find  $V_S$ ,  
then  $V_{DS} = V_{o1} - V_S$

## ■ To verify Vds:

### problem2\_dc.sp

```
.title hw4, 2: diff pair dc analysis
```

```
.options post=2 nomod
```

```
.op
```

```
.model nch nmos level=1 tox=25 vto=0.5 kp=140e-6 lambda=0.1 gamma=0.5 phi=0.6
```

```
vdd vdd 0 1.8
```

```
* set up common mode and differential inputs
```

```
vic vic 0 0.9
```

```
vid vid 0 0
```

```
e1 vi1 vic vid 0 0.5
```

```
e2 vi2 vic vid 0 -0.5
```

VCVS: Exxx n+ n- in+ in- gain  
 $V(n+, n-) = V(in+, in-) * gain$

```
rd1 vdd vo1 100k
```



```
rd2 vdd vo2 100k
```

```
m1 vo1 vi1 s 0 nch w=10u l=1u
```

```
m2 vo2 vi2 s 0 nch w=10u l=1u
```

```
iss s 0 14u
```

```
.end
```



## ■ Check .lis file:


```
**** mosfets

subckt
element 0:m1      0:m2
model    0:nch    0:nch
region   Saturati Saturati
id       7.0000u  7.0000u
ibs      -2.3459f -2.3459f
ibd      -11.0000f -11.0000f
vgs      665.4148m 665.4148m
vds      865.4147m 865.4147m
vbs      -234.5852m -234.5852m
vth      569.4798m 569.4798m
vdsat    95.9349m 95.9349m
vod      95.9349m 95.9349m
beta     1.5212m  1.5212m
gam eff  500.0000m 500.0000m
gm       145.9322u 145.9322u
gds      644.2459n 644.2459n
gmb      39.9352u 39.9352u
cdtot    159.3822a 159.3822a
cgtot    97.3634f 97.3634f
cstot    92.0843f 92.0843f
cbtot    5.1197f 5.1197f
cgs      92.0843f 92.0843f
cgd      159.3822a 159.3822a
```

Line: 1/168      Column: 1      Character: 85 (0x55)



## ■ Check .sw0 file in WaveViewer:



There are three breakpoints. Explain what causes them and calculate their positions by hand.

## ■ Problem 2(c), common mode:

problem2\_comm.sp

```
.title hw4, 2: diff pair common mode
```

```
.options post=2 nomod
```

```
.model nch nmos level=1 tox=25 vto=0.5 kp=140e-6 lambda=0.1 gamma=0.5 phi=0.6
```

```
vdd vdd 0 1.8
```

```
* set up common mode and differential inputs
```

```
vic vic 0 0.9
```

```
vid vid 0 0
```

```
e1 vi1 vic vid 0 0.5
```

```
e2 vi2 vic vid 0 -0.5
```

```
rd1 vdd vo1 100k
```

```
rd2 vdd vo2 100k
```

```
m1 vo1 vi1 s 0 nch w=10u l=1u
```

```
m2 vo2 vi2 s 0 nch w=10u l=1u
```

```
iss s 0 14u
```

```
* set up to measure output common mode
```

```
e3 voc1 0 vo1 0 0.5
```

```
e4 voc voc1 vo2 0 0.5
```

```
rcm voc 0 1
```



```
.probe dc voc=v(voc)
```

```
.op
```


```
.dc vic 0 1.8 0.01
```

```
.tf v(voc) vic
```

```
.end
```



## Check .sw0 file in WaveViewer:



There is no breakpoint.




- Problem 2, (d)-(f):

$$A_{dm} = -g_m (R_D // r_o) \approx -g_m R_D$$

$$A_{cm} = \frac{g_m R_D}{1 + (1 + \chi)g_m (2r_o)}$$

$$R_{od} = 2(R_D // r_o)$$

## ■ Problem 2, (d)-(f): HSPICE verification




The screenshot shows a terminal window with the title bar "/home/ece304/Hw\_4/problem2\_diff.lis - ece304@ece3.ece.arizona.edu". The main content area displays the following text:

```
****      small-signal transfer characteristics|  
  
v(vol,vo2)/vid          = -13.7100 ← Aod  
input resistance at      vid          = 1.000e+20  
output resistance at v(vol,vo2)      = 187.8949k ← Rod
```

At the bottom of the window, the status bar shows "Line: 491/551", "Column: 48", and "Character: 13 (0x0D)".

problem2\_diff.lis




The screenshot shows a terminal window with the title bar "/home/ece304/Hw\_4/problem2\_comm.lis - ece304@ece3.ece.arizona.edu". The main content area displays the following text:

```
****      small-signal transfer characteristics  
  
v(voc)/vic              = -79.0254n ← Aoc  
input resistance at      vic          = 1.000e+20  
output resistance at v(voc)      = 0.
```

At the bottom of the window, the status bar shows "Line: 315/372", "Column: 60", and "Character: 13 (0x0D)".

problem2\_comm.lis

■ Problem 3, (a): calculate  $V_{DS}$




To get the same current,

$$R_{SS} = \frac{V_S}{I_{SS}} = \frac{0.231V}{14\mu A} = 16.5K\Omega$$

The calculation of  $V_{DS}$  is the same as in problem 2.

- Problem 3, (d)-(f): the calculations are also similar,  $A_{cm}$  is slightly different:



Equivalent  
Half-circuit

$$A_{dm} = -g_m (R_D // r_o)$$

$$A_{cm} = \frac{g_m R_D}{1 + (1 + \chi) g_m (2R_{SS})}$$

$$R_{od} = 2(R_D // r_o)$$



- HSPICE verification:

You only need to do following modification:

```
iss s 0 14u
```



```
rss s 0 16.5k
```

The remaining part is the same.